

Dependent Types for Distributed Arrays

Wouter Swierstra and Thorsten Altenkirch

University of Nottingham
{wss,txa}@cs.nott.ac.uk

Abstract. Locality-aware algorithms over distributed arrays can be very difficult to write. Yet such algorithms are becoming more and more important as desktop machines boast more and more processors. We show how a dependently-typed programming language can help develop such algorithms by hosting a domain-specific embedded type system that ensures every well-typed program will only ever access local data. Such static guarantees can help catch programming errors early on in the development cycle and maximise the potential speedup that multicore machines offer. At the same time, the functional specification of effects we provide facilitates the testing of and reasoning about algorithms on distributed arrays.

1 Introduction

Computer processors are not becoming significantly faster. To satisfy the demand for more and more computational power, manufacturers are now assembling computers with multiple microprocessors. It is hard to exaggerate the impact this will have on software development: tomorrow’s programming languages must embrace parallel programming on multicore machines.

Researchers have proposed several new languages to maximise the potential speedup that multicore processors offer [2, 7–9]. Although all these languages are different, they share the central notion of a *distributed array*, where the elements of an array may be distributed over separate processors or even over separate machines. To write efficient code, programmers must ensure that processors only access *local* parts of a distributed array—it is much faster to access data stored locally than remote data on another core.

When writing such locality-aware algorithms it is all too easy to make subtle mistakes. Programming languages such as X10 require all arrays operations to be local [9]. Any attempt to access non-local data results in an exception. To preclude such errors, Grothoff *et al.* have designed a type system for a small core language resembling X10 that is specifically designed to guarantee that programs only access local parts of a distributed array [10]. Yet it is somehow disappointing that these static guarantees require a custom-built type system to solve a very specific problem.

In this paper, we explore an alternative avenue of research. Designing and implementing a type system from scratch is a lot of work. New type systems typically require extensive proofs of various soundness, completeness, principle

typing, and decidability theorems. Instead, we show how to tailor a sufficiently expressive type system to enforce certain properties—resulting in a *domain-specific embedded type system*. We no longer need to prove any meta-theoretical results, but immediately inherit all the desirable properties of our host type system. Functional programmers have studied domain-specific embedded languages for years [12]; the time is ripe to take these ideas one step further.

In previous work [19], we described a pure specification of several parts of the IO monad, the interface between pure functional languages such as Haskell [17] and the ‘real world.’ By providing functional, executable specifications we can test, debug, and reason about impure programs as if they were pure. When we now shift to a dependently-typed programming language, we can show how our specifications can provide stronger static guarantees about our programs. To this end, we make several novel contributions:

- We begin by giving a pure specification of arrays (Section 3). Due to our rich type system, the specification is *total*: there is no way to access unallocated memory; there are no ‘array index out of bounds’ exceptions. As a result, these specifications can not only be used to *program* with, but also facilitate *formal proofs* about array algorithms.
- Distributed arrays pose more of a challenge (Section 4). Not only do we attend to locality constraints, but we must also accommodate place-shifting operators. The pure specification we present is, once again, executable and total: it can be interpreted both as a domain-specific embedded language for writing algorithms on distributed arrays and as an executable denotational model for specifying and proving properties of such algorithms.
- Finally, we demonstrate how programmers may write their own locality-aware control structures using the primitives we have defined. We also implement several combinators that enable programmers to define their own distributions, specifying how large arrays must be distributed over different processors. We conclude by discussing how our specifications may be extended to cope with multidimensional arrays with more complex regions (Section 5).

Throughout this paper, we will use the dependently-typed programming language Agda [1, 15] as a vehicle of explanation. In fact using lhs2TeX [13], the sources of this paper generate an Agda program that can be compiled and executed.¹ We will briefly introduce the syntax by means of several examples, as it may be unfamiliar to many readers.

2 An overview of Agda

Data types in Agda can be defined using a similar syntax to that for Generalized Algebraic Data Types, or GADTs, in Haskell [?]. For example, consider the following definition of the natural numbers.

¹ The resulting code is available from the first author’s website.

data *Nat* : \star **where**

Zero : *Nat*

Succ : *Nat* \rightarrow *Nat*

There is one important difference with Haskell. We must explicitly state the *kind* of the data type that we are introducing; in particular, the declaration *Nat* : \star states that *Nat* is a base type.

We can define functions by pattern matching and recursion, just as in any other functional language. To define addition of natural numbers, for instance, we could write:

$_ + _ : \text{Nat} \rightarrow \text{Nat} \rightarrow \text{Nat}$
Zero + *m* = *m*
Succ *n* + *m* = *Succ* (*n* + *m*)

Note that Agda uses underscores to denote the positions of arguments when defining new operators.

Polymorphic lists are slightly more interesting than natural numbers:

data *List* (*a* : \star) : \star **where**

Nil : *List* *a*

Cons : *a* \rightarrow *List* *a* \rightarrow *List* *a*

To uniformly parameterise a data type, we can write additional arguments to the left of the copula. In this case, we add (*a* : \star) to our data type declaration to state that lists are type *constructors*, parameterised over a type variable *a* that has kind \star .

Just as we defined addition for natural numbers, we can define an operator that appends one list to another:

$append : (a : \star) \rightarrow List\ a \rightarrow List\ a \rightarrow List\ a$
append *a Nil* *ys* = *ys*
append *a (Cons* *x xs)* *ys* = *Cons* *x (append* *a xs* *ys)*

The *append* function is polymorphic. In Agda, such polymorphism can be introduced via the *dependent function space*, written (*x* : *a*) \rightarrow *y*, where the variable *x* may occur in the type *y*. This particular example of the dependent function space is not terribly interesting: it merely expresses parametric polymorphism. Later we will encounter more interesting examples, where *types* depend on *values*.

One drawback of using the dependent function space for such parametric polymorphism, is that we must explicitly instantiate polymorphic functions. For example, the recursive call to *append* in the *Cons* case takes a type as its first argument. Fortunately, Agda allows us to mark certain arguments as *implicit*. Using implicit arguments, we can define the $\#$ operator just as in any other functional language:

$_ \# _ : \{a : \star\} \rightarrow List\ a \rightarrow List\ a \rightarrow List\ a$
Nil $\#$ *ys* = *ys*
(Cons *x xs)* $\#$ *ys* = *Cons* *x (xs* $\#$ *ys)*

Arguments enclosed in curly brackets, such as $\{a : \star\}$, are implicit: we do not write a to the left of the equals sign and do not pass a type argument when we call the make a recursive call. The Agda type checker will automatically instantiate this function whenever we call it, much in the same way as type variables are instantiated in Haskell.

Besides polymorphic data types, Agda also supports *indexed families*. Like Haskell’s GADTs, indexed families allow a data type’s constructors to have different codomains. Indexed families, however, are more general as they also capture data types that are indexed by *values* instead of types. For example, we can define the family of finite types:

```
data Fin : Nat →  $\star$  where
  Fz : {n : Nat} → Fin (Succ n)
  Fs : {n : Nat} → Fin n → Fin (Succ n)
```

The type $Fin\ n$ corresponds to a finite type with n distinct values. For example, $Fin\ 1$ is isomorphic to the unit type; $Fin\ 2$ is isomorphic to $Bool$. Note that the argument n is left implicit in both the constructors of Fin . From the types of these constructors, it is easy to see that $Fin\ 0$ is uninhabited. For every n , the Fs constructor embeds $Fin\ n$ into $Fin\ (Succ\ n)$; the Fz constructor, on the other hand, adds a single new element to $Fin\ (Succ\ n)$ that was not in $Fin\ n$. By induction it is easy to see that $Fin\ n$ does indeed have n elements.

Agda has many other features, such as records and a rich module system, that we will hardly use in this paper. Although there are a few more concepts we will need, we will discuss them as they pop up in later sections.

3 Mutable arrays

With this brief Agda tutorial under our belt, we can start our specification of mutable arrays. We will specify three different operations on arrays: the creation of new arrays; reading from an array; and updating a value stored in an array. Before we can define the behaviour of these operations, however, we need to introduce several data types to describe the layout and contents of memory. Using these data types, we can proceed by defining an IO type that captures the syntax of array operations. Finally, we define a run function that describes how the array operations affect the heap, assigning semantics to our syntax. This semantics can be used to simulate and reason about computations on mutable arrays in a pure functional language. When compiled, however, these operations should be replaced by their more efficient, low-level counterparts.

To keep things simple, we will only work with flat arrays storing natural numbers. This is, of course, a drastic oversimplification. The techniques we present here, however, can be adapted to cover multidimensional arrays that may store different types of data (Section 5).

To avoid confusion between numbers denoting the size of an array and the data stored in an array, we introduce the $Data$ type synonym. Throughout the

rest of this paper, we will use *Data* to refer to the data stored in arrays; the *Nat* type will always refer to the size of an array.

$$\begin{aligned} \mathit{Data} &: \star \\ \mathit{Data} &= \mathit{Nat} \end{aligned}$$

Using the *Fin* type, we can give a functional specification of arrays of a fixed size by mapping every index to the corresponding datum.

$$\begin{aligned} \mathit{Array} &: \mathit{Nat} \rightarrow \star \\ \mathit{Array} \ n &= \mathit{Fin} \ n \rightarrow \mathit{Data} \end{aligned}$$

How should we represent the heap? We need to be a bit careful—as the heap will store arrays of different sizes, its type should make it clear how many arrays it stores and how large each array is. To accomplish this, we begin by introducing a data type representing the *shape* of the heap:

$$\begin{aligned} \mathit{Shape} &: \star \\ \mathit{Shape} &= \mathit{List} \ \mathit{Nat} \end{aligned}$$

The *Shape* of the heap is simply a list of natural numbers, representing the size of the arrays stored in memory.

We can now define a *Heap* data type that is indexed by a *Shape*. The *Empty* constructor corresponds to an empty heap; the *Alloc* constructor adds an array of size *n* to any heap of shape *ns* to build a larger heap with the larger layout *Cons n ns*.

$$\begin{aligned} \mathbf{data} \ \mathit{Heap} &: \mathit{Shape} \rightarrow \star \ \mathbf{where} \\ \mathit{Empty} &: \mathit{Heap} \ \mathit{Nil} \\ \mathit{Alloc} &: \{ n : \mathit{Nat} \} \rightarrow \{ ns : \mathit{Shape} \} \rightarrow \\ &\quad \mathit{Array} \ n \rightarrow \mathit{Heap} \ ns \rightarrow \mathit{Heap} \ (\mathit{Cons} \ n \ ns) \end{aligned}$$

Finally, we will want to model references, denoting locations in the heap. A value of type *Loc n ns* corresponds to a reference to an array of size *n* in a heap with shape *ns*. The *Loc* data type shares a great deal of structure with the *Fin* type. Every non-empty heap has a *Top* reference; we can weaken any existing reference to denote the same location in a larger heap using the *Pop* constructor.

$$\begin{aligned} \mathbf{data} \ \mathit{Loc} &: \mathit{Nat} \rightarrow \mathit{Shape} \rightarrow \star \ \mathbf{where} \\ \mathit{Top} &: \{ n : \mathit{Nat} \} \rightarrow \{ ns : \mathit{Shape} \} \rightarrow \mathit{Loc} \ n \ (\mathit{Cons} \ n \ ns) \\ \mathit{Pop} &: \mathbf{forall} \ \{ n \ k \ ns \} \rightarrow \mathit{Loc} \ n \ ns \rightarrow \mathit{Loc} \ n \ (\mathit{Cons} \ k \ ns) \end{aligned}$$

Note that in the type signature of the *Pop* constructor, we omit the types of three implicit arguments and quantify over them using the **forall** keyword. Alternatively, we could also have written the more verbose:

$$\begin{aligned} \mathit{Pop} &: \{ n : \mathit{Nat} \} \rightarrow \{ k : \mathit{Nat} \} \rightarrow \{ ns : \mathit{Shape} \} \rightarrow \\ &\quad \mathit{Loc} \ n \ ns \rightarrow \mathit{Loc} \ n \ (\mathit{Cons} \ k \ ns) \end{aligned}$$

When we use the **forall** keyword, the types of n , k , and ns are inferred from the rest of the signature by the Agda type checker. We will use this shorthand notation to make large type signatures somewhat more legible.

With these data types in place, we can define a data type capturing the syntax of the permissible operations on arrays. Crucially, the *IO* type is indexed by *two* shapes: a value of type $IO\ a\ ns\ ms$ denotes a computation that takes a heap of shape ns to a heap of shape ms and returns a result of type a . This pattern of indexing operations by an initial and final ‘state’ is a common pattern in dependently-typed programming [14].

```

data IO (a : ★) : Shape → Shape → ★ where
  Return : {ns : Shape} → a → IO a ns ns
  Write : forall {n ns ms} →
    Loc n ns → Fin n → Data → IO a ns ms → IO a ns ms
  Read : forall {n ns ms} →
    Loc n ns → Fin n → (Data → IO a ns ms) → IO a ns ms
  New : forall {ns ms} →
    (n : Nat) → (Loc n (Cons n ns) → IO a (Cons n ns) ms) →
    IO a ns ms

```

The *IO* type has four constructors. The *Return* constructor returns a pure value of type a without modifying the heap. The *Write* constructor takes four arguments: the location of an array of size n ; an index in that array; the value to write at that index; and the rest of the computation. Similarly, reading from an array requires a reference to an array and a suitable index. Instead of requiring the data to be written, however, the last argument of the *Read* constructor may refer to data that has been read. Finally, the *New* constructor actually changes the size of the heap. Given a number n , it allocates an array of size n on the heap; the second argument of *New* may then use this fresh reference to continue the computation in a larger heap.

The *IO* data type is a *parameterised monad* [4]—that is, a monad with *return* and *bind* operators that satisfy certain coherence conditions regarding how they may affect the heap.

```

return : {a : ★} → {ns : Shape} → a → IO a ns ns
return x = Return x
_ >>= _ : forall {a b ns ms ks} →
  IO a ns ms → (a → IO b ms ks) → IO b ns ks
Return x >>= f = f x
Write a i x wr >>= f = Write a i x (wr >>= f)
Read a i rd >>= f = Read a i (λx → rd x >>= f)
New n io >>= f = New n (λa → io a >>= f)

```

The *return* of the *IO* data type lifts a pure value into a computation that can run on a heap of any size. Furthermore, *return* does not modify the shape of the heap. The *bind* operator, $\gg=$, can be used to compose monadic computations.

To sequence two computations, the heap resulting from the first computation must be a suitable starting point for the second computation. This condition is ensured by the type of the bind operator:

To actually write programs using these arrays, we need to introduce a smart constructor for the *IO* data type. For example, we could define the *readArray* function as follows:

$$\begin{aligned} \text{readArray} &: \mathbf{forall} \{n \ ns\} \rightarrow \text{Loc } n \ ns \rightarrow \text{Fin } n \rightarrow \text{IO Data } ns \ ns \\ \text{readArray } a \ i &= \text{Read } a \ i \ \text{Return} \end{aligned}$$

There is a slight problem with this definition. As we allocate new memory, the size of the heap changes; correspondingly, we must explicitly modify any existing pointers to denote locations in a larger heap. We can achieve this by revising the above definition slightly, ensuring that an additional *Pop* constructor is wrapped around any existing references. For the purpose of this paper, however, we will ignore this technicality: the above definition will suffice.

Denotational model

We have described the syntax of array computations using the *IO* data type, we have not specified how these computation *behave*. Recall that we can model arrays as functions from indices to natural numbers:

$$\begin{aligned} \text{Array} &: \text{Nat} \rightarrow \star \\ \text{Array } n &= \text{Fin } n \rightarrow \text{Data} \end{aligned}$$

Before specifying the behaviour of *IO* computations, we define a several auxiliary functions to update an array and lookup a value stored in an array.

$$\begin{aligned} \text{lookup} &: \mathbf{forall} \{n \ ns\} \rightarrow (l : \text{Loc } n \ ns) \rightarrow \text{Fin } n \rightarrow \text{Heap } ns \rightarrow \text{Data} \\ \text{lookup } \text{Top} \quad i \ (\text{Alloc } a \ _) &= a \ i \\ \text{lookup } (\text{Pop } k) \ i \ (\text{Alloc } _ \ h) &= \text{lookup } k \ i \ h \end{aligned}$$

The *lookup* function takes a reference to an array *l*, an index *i* in the array at location *l*, and a heap, and returns the value stored in the array at index *i*. It dereferences *l*, resulting in a function of type *Fin n → Data*; the value stored at index *i* is the result of applying this function to *i*.

Next, we define a pair of functions to update the contents of an array.

$$\begin{aligned} \text{updateArray} &: \{n : \text{Nat}\} \rightarrow \text{Fin } n \rightarrow \text{Data} \rightarrow \text{Array } n \rightarrow \text{Array } n \\ \text{updateArray } i \ d \ a &= \lambda j \rightarrow \mathbf{if} \ i \equiv j \ \mathbf{then} \ d \ \mathbf{else} \ a \ j \\ \text{updateHeap} &: \mathbf{forall} \{n \ ns\} \rightarrow \\ &\quad \text{Loc } n \ ns \rightarrow \text{Fin } n \rightarrow \text{Data} \rightarrow \text{Heap } ns \rightarrow \text{Heap } ns \\ \text{updateHeap } \text{Top} \quad i \ x \ (\text{Alloc } a \ h) &= \text{Alloc } (\text{updateArray } i \ x \ a) \ h \\ \text{updateHeap } (\text{Pop } k) \ i \ x \ (\text{Alloc } a \ h) &= \text{Alloc } a \ (\text{updateHeap } k \ i \ x \ h) \end{aligned}$$

The *updateArray* function overwrites the data stored at a single index. The *updateHeap* function updates a single index of an array stored in the heap. It

proceeds by dereferencing the location on the heap where the desired array is stored and updates it accordingly, leaving the rest of the heap unchanged.

Now we finally have all the pieces in place to assign semantics to *IO* computations. The *run* function below takes a computation of type *IO a ns ms* and an initial heap of shape *ns* as arguments, and returns a pair consisting of the result of the computation and the final heap of shape *ms*.

```

data Pair (a : ★) (b : ★) : ★ where
  pair : a → b → Pair a b

  run : forall {a ns ms} → IO a ns ms → Heap ns → Pair a (Heap ms)
  run (Return x) h      = pair x h
  run (Read a i rd) h   = run (rd (lookup a i h)) h
  run (Write a i x wr) h = run wr (updateHeap a i x h)
  run (New n io) h      = run (io Top) (Alloc (λi → Zero) h)

```

The *Return* constructor simply pairs the result and heap; in the *Read* case, we lookup the data from the heap and recurse with the same heap; for the *Write* constructor, we recurse with an appropriately modified heap; finally, when a new array is created, we extend the heap with a new array that stores *Zero* at every index, and continue recursively. Note that, by convention, the *Top* constructor always refers to the most recently created reference. Our smart constructors should add additional *Pop* constructors when new memory is allocated.

Example Using our smart constructors and the monad operators, we can now define functions that manipulate arrays. For example, the *swap* function exchanges the value stored at two indices:

```

swap : forall {n ns} → Loc n ns → Fin n → Fin n → IO () ns ns
swap a i j =
  readArray a i ≫ λvali →
  readArray a j ≫ λvalj →
  writeArray a i valj ≫
  writeArray a j vali

```

In a dependently-typed programming language such as Agda, we can prove properties of our code. For example, we may want to show that swapping the contents of any two array indices twice, leaves the heap intact :

```

swapProp : {n : Nat} {ns : Shape} →
  (l : Loc n ns) → (i : Fin n) → (j : Fin n) → (h : Heap ns) →
  (h ≡ snd (run (swap l i j ≫ swap l i j) h))

```

The proof requires a lemma about how *updateHeap* and *lookupHeap* interact and is not terribly interesting in itself. The fact that we can formalise such properties and have our proof verified by a computer is much more exciting.

4 Distributed arrays

Arrays are usually represented by a continuous block of memory. *Distributed arrays*, however, can be distributed over different *places*—where every place may correspond to a different core on a multiprocessor machine, a different machine on the same network, or any other configuration of interconnected computers.

We begin by determining the type of places, where data is stored and code is executed. Obviously, we do not want to fix the type of all possible places prematurely: you may want to execute the same program in different environments. Yet regardless of the exact number of places, there are certain operations you will always want to perform, such as iterating over all places, or checking when two places are equal.

We therefore choose to abstract over the *number* of places in the module we will define in the coming section. Agda allows modules to be parametrised as follows:

```
module DistrArray (placeCount : Nat) where
```

When we import the *DistrArray* module, we are obliged to choose the number of places. Typically, there will one place for every available processor. From this number, we can define a data type corresponding to the available places:

```
Place :  $\star$   
Place = Fin placeCount
```

The key idea underlying our model of locality-aware algorithms is to index computations by the place where they are executed. The new type declaration for the *IO* monad corresponding to operations on *distributed* arrays will become:

```
data DIO (a :  $\star$ ) : Shape  $\rightarrow$  Place  $\rightarrow$  Shape  $\rightarrow$   $\star$  where  
...
```

You may want to think of a value of type *DIO a ns p ms* as a computation that can be executed at place *p* and will take a heap of shape *ns* to a heap of shape *ms*, yielding a final value of type *a*.

We strive to ensure that any well-typed program written in the *DIO* monad will never access data that is not local. The specification of distributed arrays now poses a twofold problem: we want to ensure that the array manipulations from the previous section are ‘locality-aware,’ that is, we must somehow restrict the array indices that can be accessed from a certain place; furthermore, X10 facilitates several *place-shifting* operations that change the place where certain chunks of code are executed. As we shall see in the rest of this section, both these issues can be resolved quite naturally.

Regions, Points, and Distributed Arrays

Before we define the *DIO* monad, we need to introduce several new concepts. In what follows, we will try to stick closely to X10’s terminology for distributed

arrays. Every array is said to have a *region* associated with it. A region is a set of valid index points. A *distribution* specifies a place for every index point in a region.

Once again, we will only treat flat arrays storing natural numbers and defer any discussion about how to deal with more complicated data structures for the moment. In this simple case, a region merely determines the size of the array.

$$\begin{aligned} \textit{Region} &: \star \\ \textit{Region} &= \textit{Nat} \end{aligned}$$

As we have seen in the previous section, we can model array indices using the *Fin* data type:

$$\begin{aligned} \textit{Point} &: \textit{Region} \rightarrow \star \\ \textit{Point } n &= \textit{Fin } n \end{aligned}$$

To model distributed arrays, we now need to consider the distribution that specifies *where* this data is stored. In line with existing work [10], we assume the existence of a fixed distribution. Agda’s **postulate** expression allows us to assume the existence of a distribution, without providing its definition.

$$\begin{aligned} \mathbf{postulate} \\ \textit{distr} &: \mathbf{forall} \{ n \textit{ ns} \} \rightarrow \textit{Loc } n \textit{ ns} \rightarrow \textit{Point } n \rightarrow \textit{Place} \end{aligned}$$

At the end of this section, we will discuss how programmers can define their own distributions.

Now that we have all the required auxiliary data types, we proceed by defining the *DIO* monad.

$$\mathbf{data} \textit{DIO} (a : \star) : \textit{Shape} \rightarrow \textit{Place} \rightarrow \textit{Shape} \rightarrow \star \mathbf{where}$$

As it is a bit more complex than the data types we have seen so far, we will discuss every constructor individually.

The *Return* constructor is analogous to one we have seen previously for the *IO* monad: it lifts any pure value into the *DIO* monad.

$$\textit{Return} : \{ p : \textit{Place} \} \rightarrow \{ ns : \textit{Shape} \} \rightarrow a \rightarrow \textit{DIO } a \textit{ ns } p \textit{ ns}$$

The *Read* and *Write* operations are more interesting. Although they correspond closely to the operations we have seen in the previous section, their type now keeps track of the place where they are executed. Any read or write operation to point *pt* of an array *l* can *only* be executed at the place specified by the distribution. This invariant is enforced by the types of our constructors:

$$\begin{aligned} \textit{Read} &: \mathbf{forall} \{ n \textit{ ns } \textit{ ms} \} \rightarrow \\ &(\textit{l} : \textit{Loc } n \textit{ ns}) \rightarrow (\textit{pt} : \textit{Point } n) \rightarrow \\ &(\textit{Data} \rightarrow \textit{DIO } a \textit{ ns} (\textit{distr } \textit{l } \textit{pt}) \textit{ ms}) \rightarrow \\ &\textit{DIO } a \textit{ ns} (\textit{distr } \textit{l } \textit{pt}) \textit{ ms} \end{aligned}$$

$$\begin{aligned}
\text{Write} &: \mathbf{forall} \{ n \ ns \ ms \} \rightarrow \\
& (l : \text{Loc } n \ ns) \rightarrow (pt : \text{Point } n) \rightarrow \text{Data} \rightarrow \\
& \text{DIO } a \ ns \ (\text{distr } l \ pt) \ ms \rightarrow \\
& \text{DIO } a \ ns \ (\text{distr } l \ pt) \ ms
\end{aligned}$$

In contrast to *Read* and *Write*, new arrays can be allocated at any place.

$$\begin{aligned}
\text{New} &: \mathbf{forall} \{ p \ ns \ ms \} \rightarrow \\
& (n : \text{Nat}) \rightarrow \\
& (\text{Loc } n \ (\text{Cons } n \ ns) \rightarrow \text{DIO } a \ (\text{Cons } n \ ns) \ p \ ms) \rightarrow \\
& \text{DIO } a \ ns \ p \ ms
\end{aligned}$$

Finally, we add a constructor for the place-shifting operator *At*:

$$\begin{aligned}
\text{At} &: \mathbf{forall} \{ p \ ns \ ms \ ps \} \rightarrow \\
& (q : \text{Place}) \rightarrow \text{DIO } () \ ns \ q \ ms \rightarrow \text{DIO } a \ ms \ p \ ps \rightarrow \text{DIO } a \ ns \ p \ ps
\end{aligned}$$

The *At* operator lets us execute a computation at another place. As we will discard the result of this computation, we require it to return an element of the unit type.

We can add our smart constructors for each these operations, as we have done in the previous section. We can also show that *DIO* is indeed a parameterised monad. We have omitted the definitions of the *return* and *bind* operators for the sake of brevity:

$$\begin{aligned}
\text{return} &: \mathbf{forall} \{ ns \ a \ p \} \rightarrow a \rightarrow \text{DIO } a \ ns \ p \ ns \\
- \gg= - &: \mathbf{forall} \{ ns \ ms \ ks \ a \ b \ p \} \rightarrow \\
& \text{DIO } A \ ns \ p \ ms \rightarrow (A \rightarrow \text{DIO } B \ ms \ p \ ks) \rightarrow \text{DIO } B \ ns \ p \ ks
\end{aligned}$$

It is worth noting that the *bind* operator $\gg=$ can only be used to sequence operations at the same place.

We can use our primitives to define more interesting functions over distributed arrays. The distributed map, for example, applies a function to all the elements of a distributed array at the place where they are stored.

$$\begin{aligned}
\text{for} &: \mathbf{forall} \{ n \ ns \ p \} \rightarrow (\text{Point } n \rightarrow \text{DIO } () \ ns \ p \ ns) \rightarrow \text{DIO } () \ ns \ p \ ns \\
\text{for } \{ \text{Succ } k \} \text{ dio} &= \text{dio } Fz \gg= (\text{for } \{ k \} (\text{dio} . Fs)) \\
\text{for } \{ \text{Zero} \} \text{ dio} &= \text{return } () \\
\text{dmap} &: \mathbf{forall} \{ n \ ns \ p \} \rightarrow (\text{Data} \rightarrow \text{Data}) \rightarrow \text{Loc } n \ ns \rightarrow \text{DIO } () \ ns \ p \ ns \\
\text{dmap } f \ l = \text{for } (\lambda i &\rightarrow \text{at } (\text{distr } l \ i) (\text{readArray } l \ i \gg= \lambda x \rightarrow \\
& \text{writeArray } l \ i \ (f \ x)))
\end{aligned}$$

We use the *for* function defined above to iterate over the indices of an array. The *dmap* function illustrates how these primitives we have presented can be used to write distributed algorithms.

Denotational model

To run a computation in the *DIO* monad, we stick quite closely to the *run* function defined in the previous section. Our new *run* function, however, must be locality-aware. To that end, we parameterise the *run* function explicitly by the place where the computation is executed.

$$\begin{aligned}
\text{run} &: \mathbf{forall} \{ a \ ns \ ms \} \rightarrow \\
& (p : \text{Place}) \rightarrow \text{DIO } a \ ns \ p \ ms \rightarrow \text{Heap } ns \rightarrow \text{Pair } a \ (\text{Heap } ms) \\
\text{run } p \ (\text{Return } x) \ h &= \text{pair } x \ h \\
\text{run } \text{.}(distr \ l \ i) \ (\text{Read } l \ i \ rd) \ h &= \text{run } (distr \ l \ i) \ (rd \ (\text{lookup } l \ i \ h)) \ h \\
\text{run } \text{.}(distr \ l \ i) \ (\text{Write } l \ i \ x \ wr) \ h &= \mathbf{let} \ h' = \text{updateHeap } l \ i \ x \ h \\
& \quad \mathbf{in} \ \text{run } (distr \ l \ i) \ wr \ h' \\
\text{run } p \ (\text{New } n \ io) \ h &= \text{run } p \ (io \ Top) \ (\text{Alloc } (\lambda i \rightarrow Zero) \ h) \\
\text{run } p \ (\text{At } q \ io1 \ io2) \ h &= \text{run } p \ io2 \ (\text{snd } (\text{run } q \ io1 \ h))
\end{aligned}$$

Now we can see that the *Read* and *Write* operations may not be executed at *any* place. Recall that the *Read* and *Write* constructors both return computations at the place *distr l i*. When we pattern match on a *Read* or *Write*, we know exactly what the place argument of the *run* function must be. Correspondingly, we do not pattern match on the place argument—we know that the place can only be *distr l i*. Agda’s syntax allows us to prefix expressions by a single period, provided we know that there is only one possible value an argument may take. This may be unfamiliar to many functional programmers who are used to thinking of patterns being built-up from variables and constructors—*distr l i* is an expression, not a pattern! The situation is somewhat similar to pattern matching on GADTs in Haskell, which introduces equalities between *types*. The *DIO* monad, however, is indexed by values. As a result, pattern matching in the presence of dependent types may introduce equalities between *values*.

The other difference with respect to the previous *run* function, is the new case for the *At* constructor. In that case, we sequence the two computations *io1* and *io2*. To do so, we first execute the *io1* at *q*, but discard its result; we continue executing the second computation *io2* with the heap resulting from the execution of *io1* at the location *p*. Conform to previous proposals [?], we have assumed that *io1* and *io2* are performed synchronously—executing *io1* before continuing with the rest of the computation. Using techniques to model concurrency that we have presented previously [19], we believe we could give a more refined treatment of the X10’s *globally asynchronous/locally synchronous* semantics and provide specifications for X10’s *clocks*, *finish*, and *force* constructs.

Other operations

Using the place-shifting operator *at*, we can define several other operations to manipulate places and regions. With our first-class distribution and definition of *Place*, we believe there is no need to define more primitive operations. We show how two new control structures, *forallplaces* and *ateach*, can be defined in terms of the functions we have seen so far.

The *forallplaces* operation executes its argument computation at all available places. We define it using the *for* function to iterate over all places. The *ateach* function, on the other hand, is a generalisation of the distributed map operation. It iterates over an array, executing its argument operation once for every index of the array, at the place where that index is stored.

$$\begin{aligned}
\text{forallplaces} &: \mathbf{forall} \{p \ ns\} \rightarrow \\
& ((q : \text{Place}) \rightarrow \text{DIO } () \ ns \ q \ ns) \rightarrow \text{DIO } () \ ns \ p \ ns \\
\text{forallplaces } io &= \text{for } (\lambda i \rightarrow \text{at } i \ (io \ i)) \\
\text{ateach} &: \mathbf{forall} \{n \ ns \ p\} \rightarrow \\
& (l : \text{Loc } n \ ns) \rightarrow ((pt : \text{Point } n) \rightarrow \text{DIO } () \ ns \ (\text{distr } l \ pt) \ ns) \rightarrow \\
& \text{DIO } () \ ns \ p \ ns \\
\text{ateach } l \ io &= \text{for } (\lambda i \rightarrow \text{at } (\text{distr } l \ i) \ (io \ i))
\end{aligned}$$

Defining distributions

Throughout this section we have assumed the existence of a distribution, specifying how an array is distributed over the available places. Several built-in distributions are provided by X10, some of which we will now define. For the moment, we focus on defining a distribution of a single array, that is, functions of type $\text{Fin } n \rightarrow \text{Place}$.

There are two atomic distributions: the constant distribution maps all the points of an array to a single place; the *unique* distribution maps the i -th index of an array to the i -th place.

$$\begin{aligned}
\text{constDistr} &: \{n : \text{Nat}\} \rightarrow \text{Place} \rightarrow \text{Point } n \rightarrow \text{Place} \\
\text{constDistr } p &= \lambda i \rightarrow p \\
\text{unique} &: \text{Point } \text{placeCount} \rightarrow \text{Place} \\
\text{unique } i &= i
\end{aligned}$$

We can compose any two such distributions, to form a larger distribution:

$$\begin{aligned}
\text{compose} &: \mathbf{forall} \{n \ m\} \rightarrow \\
& (\text{Point } n \rightarrow \text{Place}) \rightarrow (\text{Point } m \rightarrow \text{Place}) \rightarrow (\text{Point } (n + m) \rightarrow \text{Place})
\end{aligned}$$

Although the definition of *compose* is a bit tricky, there is a particularly elegant definition in the literature using *views* [?]. Of course, we can iterate the *compose* operator:

$$\begin{aligned}
\text{cycle} &: \{n : \text{Nat}\} \rightarrow \\
& (k : \text{Nat}) \rightarrow (\text{Point } n \rightarrow \text{Place}) \rightarrow (\text{Point } (k * n) \rightarrow \text{Place}) \\
\text{cycle } (\text{Succ } k) \ f \ i &= \text{compose } f \ (\text{cycle } k \ f) \ i
\end{aligned}$$

In the case where k is equal to *Zero*, we need to define a function of type $\text{Fin } \text{Zero} \rightarrow \text{Place}$. As $\text{Fin } \text{Zero}$ has no inhabitants, this function will never be applied and we may omit the definition accordingly.

Using similar combinators, we can define distributions over all arrays, exploiting the obvious similarity between *Loc* and *Fin*. There are several other distributions supported by X10, that can be implemented along the same lines.

5 Discussion

Using a dependently-typed host language, we have shown how to implement a domain-specific library for distributed arrays, together with an embedded type system that guarantees all array access operations are both safe and local. We have provided semantics for our library in the form of a total, functional specification. Although our semantics may not take the form of deduction rules, they are no less precise or concise. Besides these functional specifications are both executable and amenable to computer-aided formal verification. More generally, we hope that this approach can be extended to other domains: a dependently-typed language accommodates domain specific libraries with their own embedded type systems.

Having said this, there are clearly several serious limitations of this work as it stands. First and foremost, we have assumed that every array only stores natural numbers, disallowing more complex structures such as multidimensional arrays. This can be easily fixed by defining a more elaborate *Shape* data type. In its most general form, we could choose our *Shape* data type as a list of types; a heap then corresponds to a list of values of the right type.² We decided to restrict ourselves to this more simple case for the purpose of presentation. We also believe that there is no fundamental obstacle preventing us from incorporating the rich region calculus offered by X10 in the same fashion.

Secondly, we have not discussed how code in the *IO* or *DIO* monad is actually compiled. To actually exploit the locality information our types carry, we would need to customise the existing Agda to Haskell compiler. The ongoing effort to support data parallelism in Haskell [5, 6] could hopefully provide us with a most welcome foothold.

Furthermore, we have assumed a fixed distribution. We would like to investigate a more flexible approach, allowing distributions to be associated with arrays as they are created, rather than require a global distribution to be chosen before a program is executed.

Finally, there are many features of X10 that we have not discussed here at all. Most notably, we have refrained from modelling many of X10's constructs that enable asynchronous communication between locations, even though we would like to do so in the future. Nonetheless, we believe this paper demonstrates the viability of our approach and provides a first stepping-stone for such further research.

Acknowledgements We wish to express our gratitude to Jens Palsberg for our interesting discussions; to Ulf Norell for his fantastic new incarnation of Agda; and to Mauro Jaskelioff, Nicolas Oury, and Liyang HU for their helpful comments on a draft version of this paper.

² There are some technical details involving ‘size problems’ that are beyond the scope of this paper. The standard technique of introducing a universe, closed under natural numbers and arrays, should resolve these issues.

References

1. Agda 2. <http://appserv.cs.chalmers.se/users/ulfn/wiki/agda.php>.
2. Eric Allen, David Chase, Victor Luchangco, Jan-Willem Maessen, Sukyoung Ryu, Guy L. Steele Jr., and Sam Tobin-Hochstadt. The Fortress language specification. Technical report, Sun Microsystems, Inc., 2005.
3. Thorsten Altenkirch, Conor McBride, and Peter Morris. Generic programming with dependent types. Draft lecture notes for the Summer School on Generic Programming, August 2006.
4. Robert Atkey. Parameterised notions of computation. In *Proceedings of the Workshop on Mathematically Structured Functional Programming*, 2006.
5. Manuel M.T. Chakravarty, Gabriele Keller, Roman Lechtchinsky, and Wolf Pfannenstiel. Nepal – Nested Data-Parallelism in Haskell. In *Euro-Par 2001: Parallel Processing, 7th International Euro-Par Conference*, volume LNCS 2150, 2001.
6. Manuel M.T. Chakravarty, Roman Leshchinskiy, Simon Peyton Jones, Gabriele Keller, and Simon Marlow. Data Parallel Haskell: a status report. *Proceedings of the 2007 Workshop on Declarative Aspects of Multicore Programming*, 2007.
7. Brad Chamberlain, Steve Deitz, Mary Beth Hribar, and Wayne Wong. Chapel. Technical report, Cray Inc., 2005.
8. Bradford L. Chamberlain, Sung-Eun Choi, E. Christopher Lewis, Calvin Lin, Lawrence Snyder, and Derrick Weathersby. ZPL: A machine independent programming language for parallel computers. *Software Engineering*, 26(3):197–211, 2000.
9. Philippe Charles, Christian Grothoff, Vijay Saraswat, Christopher Donawa, Allan Kielstra, Kemal Ebcioglu, Christoph von Praun, and Vivek Sarkar. X10: an object-oriented approach to non-uniform cluster computing. In *OOPSLA '05: Proceedings of the 20th Annual ACM SIGPLAN Conference on Object Oriented Programming, Systems, Languages, and Applications*, pages 519–538, 2005.
10. Christian Grothoff, Jens Palsberg, and Vijay Saraswat. Safe arrays via regions and dependent types. Submitted for publication.
11. Paul N. Hilfinger, Dan Oscar Bonachea, Kaushik Datta, David Gay, Susan L. Graham, Benjamin Robert Liblit, Geoffrey Pike, Jimmy Zhigang Su, and Katherine A. Yelick. Titanium language reference manual, version 2.19. Technical Report UCB/EECS-2005-15, EECS Department, University of California, Berkeley, Nov 2005.
12. Paul Hudak. Building domain-specific embedded languages. *ACM Computing Surveys*, 28, 1996.
13. Andres Löf. lhs2tex. <http://people.cs.uu.nl/andres/lhs2tex/>.
14. James McKinna and Joel Wright. A type-correct, stack-safe, provably correct, expression compiler in Epigram. Submitted to the Journal of Functional Programming, 2006.
15. Ulf Norell. *Towards a practical programming language based on dependent type theory*. PhD thesis, Chalmers University of Technology, 2007.
16. Robert W. Numrich and John Reid. Co-array fortran for parallel programming. *SIGPLAN Fortran Forum*, 17(2):1–31, 1998.
17. Simon Peyton Jones, editor. *Haskell 98 Language and Libraries – The Revised Report*. Cambridge University Press, 2003.
18. Herb Sutter. The free lunch is over. *Dr. Dobbs's Journal*, 30(3), 2005.
19. Wouter Swierstra and Thorsten Altenkirch. Beauty in the beast: a functional semantics of the awkward squad. In *Proceedings of the ACM SIGPLAN Haskell Workshop*, 2007.